/home/m262f99/KOEPF/worksheetsV.4/hw4ansm.mws

Math 262a, Fall 1999, Glenn Tesler
Homework 4

```
> read ‘hsum.mpl‘;
```

Copyright 1998 Wolfram Koepf, Konrad-Zuse-Zentrum Berlin

Problem 1

First of all, the computer can generate the recurrence that is given to you in the problem:

```
> rec_cer := sumrecursion(2^k * n/(n-k) *
  binomial(n-k,2*k),                k=0..n/3, f(n),
  certificate=true);
```

$$rec\_cer := \left[-f(n+3) + 2\, f(n+2) - f(n+1) + 2\, f(n) = 0,\ 2\, \frac{(-n+k)\,(2\,k-1)\,k}{(-n+3\,k-1)\,(3\,k-2-n)\,(-n+3\,k-3)} \right]$$

To recover what’s on the problem, replace s(n+i) by F(n+i,k) in the recurrence (first return value); and the second return value is R(n,k). Multiply it by F(n,k) to get G(n,k), then make the right side be G(n,k+1)-G(n,k).

```
> factor(-E^3+2*E^2-E+2);
```

$$-(E-2)\,(E^2+1)$$

so evidentally the left side of the recursion has an overall minus sign from what’s given

```
> F1 := 2^k * n/(n-k) * binomial(n-k,2*k);
```

$$F1 := \frac{2^k\, n \operatorname{binomial}(n-k,\, 2\,k)}{n-k}$$

```
> R1 := rec_cer[2];
  G1 := F1*R1;
```

$$R1 := 2\, \frac{(-n+k)\,(2\,k-1)\,k}{(-n+3\,k-1)\,(3\,k-2-n)\,(-n+3\,k-3)}$$

$$G1 := 2\, \frac{2^k\, n \operatorname{binomial}(n-k,\, 2\,k)\,(-n+k)\,(2\,k-1)\,k}{(n-k)\,(-n+3\,k-1)\,(3\,k-2-n)\,(-n+3\,k-3)}$$

```
> simplify(-2^k*n/(n-3*k+3) * binomial(n-k,2*k-2) / G1);
```

$$-\frac{1}{2}\, \frac{(3\,k-2-n)\,(-n+3\,k-1) \operatorname{binomial}(n-k,\, 2\,k-2)}{k\,(2\,k-1) \operatorname{binomial}(n-k,\, 2\,k)}$$

```
> simpcomb(");
```

$$-1$$

so the right side has a negative also. So Koepf's program deduced something equivalent to what's stated. Now verify it.

Problem 1a

```
> lh := subs('f(n+i)=subs(n=n+i,F1)'$i=0..3,
  lhs(rec_cer[1]));
```

$$lh := -\frac{2^k (n+3) \operatorname{binomial}(n+3-k, 2k)}{n+3-k} + 2\frac{2^k (n+2) \operatorname{binomial}(n+2-k, 2k)}{n+2-k} - \frac{2^k (n+1) \operatorname{binomial}(n-k+1, 2k)}{n-k+1} + 2\frac{2^k n \operatorname{binomial}(n-k, 2k)}{n-k}$$

```
> rh := subs(k=k+1,G1)-G1;
```

$$rh := 2\frac{2^{(k+1)} n \operatorname{binomial}(n-k-1, 2k+2)(-n+k+1)(2k+1)(k+1)}{(n-k-1)(-n+3k+2)(3k+1-n)(-n+3k)} - 2\frac{2^k n \operatorname{binomial}(n-k, 2k)(-n+k)(2k-1)k}{(n-k)(-n+3k-1)(3k-2-n)(-n+3k-3)}$$

```
> simplify(lh/F1);
```

$$2\frac{25k^3 - 25nk^2 - 53k^2 + 9n^2k + 35nk + 33k - n^3 - 11n - 6 - 6n^2}{(-n+3k-1)(3k-2-n)(-n+3k-3)}$$

```
> simplify(rh/F1);
```

$$2\frac{25k^3 - 25nk^2 - 53k^2 + 9n^2k + 35nk + 33k - n^3 - 11n - 6 - 6n^2}{(-n+3k-1)(3k-2-n)(-n+3k-3)}$$

```
> "-"";
```

$$0$$

The verification is complete.

Problem 1b

The bounds k=0..floor(n/3) aren't natural: the denominator n-k cancels off the n-k from (n-k)! in the numerator, giving

```
> F1b := simpcomb(F1);
```

$$F1b := \frac{\Gamma(n-k)\, n\, 2^k}{\Gamma(2k+1)\,\Gamma(n-3k+1)}$$

and the k-support at n is k={0,1,...,floor(n/3), n,n+1,n+2,...}; for example,

```
> '' F1'(10,kk)=limit(subs(n=10,F1b),k=kk)'$kk=-3..13;
```

$$F1(10,-3)=0,\ F1(10,-2)=0,\ F1(10,-1)=0,\ F1(10,0)=1,\ F1(10,1)=80,$$
$$F1(10,2)=350,\ F1(10,3)=80,\ F1(10,4)=0,\ F1(10,5)=0,\ F1(10,6)=0,$$
$$F1(10,7)=0,\ F1(10,8)=0,\ F1(10,9)=0,\ F1(10,10)=-1536,$$
$$F1(10,11)=-61440,\ F1(10,12)=-1536000,\ F1(10,13)=-30965760$$

but we do have F(n,k)=0 for k=floor(n/3)+1,floor(n/3)+2,...,n-1. From now on we must assume n>=2 because otherwise this k-range just given is empty, and we're assuming it's not.

The recursion involves f(n),f(n+1),f(n+2),f(n+3), and in all these the parameter is different, so the summation range is slightly different.

So take the relation

```
> -(En^2+1)*(En-2)*' F1'(n,k) = ' G1'(n,k+1)-' G1'(n,k);
```

$$-(En^2+1)(En-2)\ \mathrm{F1}(n,k) = \mathrm{G1}(n,k+1) - \mathrm{G1}(n,k)$$

and sum both sides from k=0..floor(n/3)+1. The terms on the left side still add up to

```
> lhs(rec_cer[1]);
```

$$-\mathrm{f}(n+3)+2\,\mathrm{f}(n+2)-\mathrm{f}(n+1)+2\,\mathrm{f}(n)$$

because we have added 0 to some of these. The terms on the right side add up to G1(n,floor(k/3) + 2) - G1(n,0) = 0 - 0 = 0. This proves the recursion above equals 0.

Now evaluate f(n). It's a constant coefficient recurrence with roots i, -i, 2, so

```
> fn := c1*I^n + c2*(-I)^n + c3*2^n;
```

$$fn := c1\ I^n + c2\ (-I)^n + c3\ 2^n$$

```
> ff := nn -> sum(subs(n=nn,F1),k=0..nn/3); # maple
  truncates to floor(n/3)
```

$$ff := nn \to \sum_{k=0}^{1/3\,nn} \mathrm{subs}(n=nn, F1)$$

Plug in initial conditions

```
> 'subs(n=nn,fn) = ff(nn)'$nn=2..4;
```

$$-c1 - c2 + 4\,c3 = 1,\ -I\,c1 + I\,c2 + 8\,c3 = 4,\ c1 + c2 + 16\,c3 = 9$$

```
> solve({"}, {c1,c2,c3});
```

$$\{c1 = \frac{1}{2},\ c2 = \frac{1}{2},\ c3 = \frac{1}{2}\}$$

```
> subs(",fn);
```

$$\frac{1}{2}I^n + \frac{1}{2}(-I)^n + \frac{1}{2}2^n$$

```
> fn2 := ":
```

Now check it.

```
> for nn from 0 to 10 do
        print('actual f'(nn)=ff(nn),
```

```
              'new formula for
f'(nn)=simplify(subs(n=nn,fn2)));
od;
```

$$\text{actual f}(0) = 0\text{, new formula for f}(0) = \frac{3}{2}$$

$$\text{actual f}(1) = 1\text{, new formula for f}(1) = 1$$

$$\text{actual f}(2) = 1\text{, new formula for f}(2) = 1$$

$$\text{actual f}(3) = 4\text{, new formula for f}(3) = 4$$

$$\text{actual f}(4) = 9\text{, new formula for f}(4) = 9$$

$$\text{actual f}(5) = 16\text{, new formula for f}(5) = 16$$

$$\text{actual f}(6) = 31\text{, new formula for f}(6) = 31$$

$$\text{actual f}(7) = 64\text{, new formula for f}(7) = 64$$

$$\text{actual f}(8) = 129\text{, new formula for f}(8) = 129$$

$$\text{actual f}(9) = 256\text{, new formula for f}(9) = 256$$

$$\text{actual f}(10) = 511\text{, new formula for f}(10) = 511$$

We see that it happens to hold for n=1 by accident, and indeed does not hold for n=0.

```
>
```

Problem 2

The examples shown:

```
> sumrecursion(binomial(n,k),k,s(n));
```

$$-\mathrm{s}(n+1) + 2\,\mathrm{s}(n) = 0$$

```
> zeilberger(binomial(n,k),k,s(n));
```

$$2\,\mathrm{s}(n) - \mathrm{s}(n+1) = 0$$

```
> fasenmyer(binomial(n,k),k,s(n),0);
Error, (in kfreerec) no kfree recurrence equation of order (, 0, 0, ) exists
> fasenmyer(binomial(n,k),k,s(n),1);
```

$$\mathrm{s}(n+1) - 2\,\mathrm{s}(n) = 0$$

```
> closedform(binomial(n,k),k,n);
```

$$2^n$$

```
> Closedform(binomial(n,k),k,n);
```

$$\text{Hyperterm}([1], [\ ], 2, n)$$

and now doing the given problem with each of the various algorithms:

```
> closedform(binomial(n,k)^3,k,n);
Error, (in zeilberger) algorithm finds no recurrence equation of first order
> rec1 := sumrecursion(binomial(n,k)^3,k,s(n));
```

$$rec1 := -(n+2)^2\, s(n+2) + (21\, n + 7\, n^2 + 16)\, s(n+1) + 8\, (n+1)^2\, s(n) = 0$$

```
> r1 := lhs(rec1);
```

$$r1 := -(n+2)^2\, s(n+2) + (21\, n + 7\, n^2 + 16)\, s(n+1) + 8\, (n+1)^2\, s(n)$$

Creative telescoping found a recurrence of order 2. Try that for Celine's algorithm.

```
> fasenmyer(binomial(n,k)^3,k,s(n),2);
Error, (in kfreerec) no kfree recurrence equation of order (, 2, 2, ) exists
> fasenmyer(binomial(n,k)^3,k,s(n),3);
```

$$(3\, n+4)\, (n+3)^2\, s(n+3) - 2\, (9\, n^3 + 57\, n^2 + 116\, n + 74)\, s(n+2)$$
$$-\, (3\, n+5)\, (15\, n^2 + 55\, n + 48)\, s(n+1) - 8\, (3\, n+7)\, (n+1)^2\, s(n) = 0$$

```
> rec2 := ":
> r2 := lhs(rec2);
```

$$r2 := (3\, n+4)\, (n+3)^2\, s(n+3) - 2\, (9\, n^3 + 57\, n^2 + 116\, n + 74)\, s(n+2)$$
$$-\, (3\, n+5)\, (15\, n^2 + 55\, n + 48)\, s(n+1) - 8\, (3\, n+7)\, (n+1)^2\, s(n)$$

They appear to be different!
But remember, any recursion can be "multiplied" by a polynomial in n and E to give a higher order recursion. So we could have r2 = (A+B*E)*r1, where A,B are polynomials in n; or it could be the function really satifies an order 1 recurrence, and r1, r2 are both multiples of it.

Set up the equation r2 - (A+B*E)*r1 = 0.
Collect it by s(n),s(n+1),...
The coefficients of s(n), s(n+1),... on both sides must agree, hence the coefficient on the left must equal 0. This gives a system of equations in A,B.

```
> Er1 := subs(n=n+1,r1);
```

$$Er1 := -(n+3)^2\, s(n+3) + (21\, n + 37 + 7\, (n+1)^2)\, s(n+2) + 8\, (n+2)^2\, s(n+1)$$

```
> ss := {'s(n+i)'$i=0..3};
```

$$ss := \{ s(n), s(n+1), s(n+2), s(n+3) \}$$

```
> collect(A*r1 + B*Er1 - r2,ss,factor);
```

$$8\, (n+1)^2\, (3\, n + A + 7)\, s(n) +$$
$$(8\, B\, n^2 + 32\, B\, n + 32\, B + 21\, A\, n + 7\, A\, n^2 + 16\, A + 45\, n^3 + 240\, n^2 + 419\, n + 240)$$
$$s(n+1) +$$
$$(-A\, n^2 - 4\, A\, n - 4\, A + 18\, n^3 + 114\, n^2 + 232\, n + 148 + 35\, B\, n + 44\, B + 7\, B\, n^2)$$
$$s(n+2) - (n+3)^2\, (3\, n + B + 4)\, s(n+3)$$

```
> coeffs(",ss);
```

$$8\, (n+1)^2\, (3\, n + A + 7),$$
$$8\, B\, n^2 + 32\, B\, n + 32\, B + 21\, A\, n + 7\, A\, n^2 + 16\, A + 45\, n^3 + 240\, n^2 + 419\, n + 240,$$
$$-A\, n^2 - 4\, A\, n - 4\, A + 18\, n^3 + 114\, n^2 + 232\, n + 148 + 35\, B\, n + 44\, B + 7\, B\, n^2,$$

$$-(n+3)^2 (3n+B+4)$$

```
> solve({"},{A,B});
```

$$\{B = -3n - 4, A = -3n - 7\}$$

```
> subs(", A+B*E);
```

$$-3n - 7 + (-3n - 4) E$$

```
>
```

Thus, the second recurrence is -[(3n+7)+(3n+4)E]*first recurrence.
SC is Sister Celine's algorithm, CT is Zeilberger's Creative Telescoping.
Note SC succeeds implies CT succeeds (and with a possibly smaller order recurrence), but not conversely.
SC finds a homogeneous recurrence of unknown orders in n,k for F(n,k), then converts it to a recurrence for f(n)=sum_k F(n,k).
CT finds a nonhomogeneous recurrence for f(n) of unknown order in n alone, so there may not even be a recurrence for F(n,k), or it may exist, but with higher n-order than necessary for f(n).

```
>
```

Problem 3

Koepf 7.4

Let

```
> Sum(a[i](n)*s(n+i),i=0..K)=0;
  Sum(b[i](n)*s(n+i),i=0..K)=0;
```

$$\sum_{i=0}^{K} a_i(n)\, s(n+i) = 0$$

$$\sum_{i=0}^{K} b_i(n)\, s(n+i) = 0$$

be two recurrence equations satisfied by s(n) of the same minimal order K. The first equation minus a_K(n)/b_K(n) * the second equation has order at most K-1, and thus is 0=0 because K was minimal. Thus, any two equations of minimal order are rational multiples of each other. Clearing denominators and removing all common factors, we can get the a_i(n)'s to be polynomials s.t. gcd(a_0(n),a_1(n),...,a_K(n))=1, and then all other recurrences of order K with polynomials in n as coefficients must be a polynomial multiple of this one.

```
>
>
```

Koepf 7.7

```
> sumrecursion(hyperterm([a,b],[c+m],1,k),k,s(m));
```

$$(b-c-m)(a-c-m)\,\mathrm{s}(m+1)+(c+m)(a+b-c-m)\,\mathrm{s}(m)=0$$

Koepf 7.10

```
> sumrecursion(binomial(n,k)*pochhammer(x,k)*pochhammer(y,
  n-k),k,s(n));
```

$$-\mathrm{s}(n+1)+(y+x+n)\,\mathrm{s}(n)=0$$

```
> closedform(binomial(n,k)*pochhammer(x,k)*pochhammer(y,n-
  k),k,n);
```

$$\mathrm{pochhammer}(y+x,n)$$

```
>
```

Koepf 7.11

Krawtchouk

```
> sumrecursion((-1)^n*p^n*binomial(N,n)*hyperterm([-n,-x],
  [-N],1/p,k),k,K(n));
```

$$(2+n)\,\mathrm{K}(2+n)-(2\,p-1-p\,N-n+2\,n\,p+x)\,\mathrm{K}(1+n)$$
$$+p\,(-1+p)\,(-N+n)\,\mathrm{K}(n)=0$$

```
> sumrecursion((-1)^n*p^n*binomial(N,n)*hyperterm([-n,-x],
  [-N],1/p,k),k,K(x));
```

$$p\,(x-N+1)\,\mathrm{K}(x+2)-(2\,p-1-p\,N-x+2\,x\,p+n)\,\mathrm{K}(x+1)$$
$$+(x+1)\,(-1+p)\,\mathrm{K}(x)=0$$

```
> sumrecursion((-1)^n*p^n*binomial(N,n)*hyperterm([-n,-x],
  [-N],1/p,k),k,K(N));
```

$$-(-1+p)\,(-N-2+n)\,\mathrm{K}(N+2)-(x+2\,p+n-3+p\,N-2\,N)\,\mathrm{K}(N+1)$$
$$+(x-N-1)\,\mathrm{K}(N)=0$$

```
>
```

Koepf 7.15

```
> for dd from 2 to 5 do
    print(‘recursion for d‘=dd);
    rec :=
  (sumrecursion((-1)^k*binomial(n,k)*binomial(dd*k,n),k,s(
  n)));
    print(rec);
    fn := subs(d_=dd, n -> (-d_)^n);
    print(‘Check at s‘(n)=eval(fn),‘:
  ‘,expand(eval(subs(s=fn,rec))));
```

```
od:
```

$$\textit{recursion for } d = 2$$

$$s(n+1) + 2\, s(n) = 0$$

$$\text{Check at } s(n) = (n \to (-2)^n), :\quad, 0 = 0$$

$$\textit{recursion for } d = 3$$

$$2\,(2\,n+3)\, s(n+2) + 3\,(5\,n+7)\, s(n+1) + 9\,(n+1)\, s(n) = 0$$

$$\text{Check at } s(n) = (n \to (-3)^n), :\quad, 0 = 0$$

$$\textit{recursion for } d = 4$$

$$3\,(7+3\,n)\,(4+3\,n)\,(3\,n+8)\, s(n+3) + 4\,(4+3\,n)\,(37\,n^2+180\,n+218)\, s(n+2) + 16\,(n+2)\,(33\,n^2+125\,n+107)\, s(n+1) + 64\,(7+3\,n)\,(n+2)\,(n+1)\, s(n) = 0$$

$$\text{Check at } s(n) = (n \to (-4)^n), :\quad, 0 = 0$$

$$\textit{recursion for } d = 5$$

$$8\,(2\,n+7)\,(2\,n+5)\,(4\,n+13)\,(9+4\,n)\,(4\,n+5)\,(4\,n+15)\, s(4+n) + 5\,(9\,n+31)\,(9+4\,n)\,(4\,n+5)\,(2\,n+5)\,(41\,n^2+283\,n+486)\, s(n+3) + 25\,(4\,n+5)\,(n+3)\,(1048\,n^4+12242\,n^3+52919\,n^2+100279\,n+70302)\, s(n+2) + 125\,(4\,n+13)\,(n+3)\,(n+2)\,(152\,n^3+1098\,n^2+2437\,n+1623)\, s(n+1) + 625\,(2\,n+7)\,(4\,n+13)\,(9+4\,n)\,(n+3)\,(n+2)\,(n+1)\, s(n) = 0$$

$$\text{Check at } s(n) = (n \to (-5)^n), :\quad, 0 = 0$$

Koepf 7.19(d)

```
> closedform(binomial(-1/4,k)^2*binomial(-1/4,n-k)^2,k,n);
```

$$\frac{((2\,n)!)^3}{(4^n)^3\,(n!)^6}$$

Koepf 7.19(e)

This turned out to be much more tedious than I ever imagined it would be...

```
> Fnk := hyperterm([-n,1-a-n,1-b-n],[a,b],1,k);
  Fn := hypergeom([-n,1-a-n,1-b-n],[a,b],1);
  rec := sumrecursion(Fnk,k,s(n));
```

$$Fnk := \frac{\text{pochhammer}(-n, k)\ \text{pochhammer}(1-a-n, k)\ \text{pochhammer}(1-b-n, k)}{\text{pochhammer}(a, k)\ \text{pochhammer}(b, k)\ k!}$$

$$Fn := \mathrm{hypergeom}([-n, 1-b-n, 1-a-n], [a, b], 1)$$

$$rec := (n+2\,b)\,(n+2\,a)\,(n+a+b)\,(n-1+a+b)\,\mathrm{s}(n+2) + (n+1)\,(3\,n+2+2\,a+2\,b)\,(2\,a+3\,n+2\,b)\,(3\,n+2\,b-2+2\,a)\,\mathrm{s}(n) = 0$$

It only involves s(n) and s(n+2), so it's just as easy to solve as a first order recursion would be. But Maple doesn't see that, so I aborted this computation after awhile.

```
> rsolve(rec,s(n));
Warning, computation interrupted
```

Let's transform it via n=2*N (for even n), t(N)=s(2N).
Then t(N+1)=s(2N+2) and it's a first order recursion for t(N).

```
> s_even := closedform(subs(n=2*N,Fnk),k,N);
```

$$s\_even := (2\,N)!\,\mathrm{pochhammer}\left(\frac{1}{3}+\frac{1}{3}\,b+\frac{1}{3}\,a, N\right)\mathrm{pochhammer}\left(\frac{1}{3}\,a+\frac{1}{3}\,b, N\right)\mathrm{pochhammer}\left(-\frac{1}{3}+\frac{1}{3}\,b+\frac{1}{3}\,a, N\right)(\text{-}27)^N \Big/ \left(4^N\,\mathrm{pochhammer}(b, N)\,\mathrm{pochhammer}(a, N)\,\mathrm{pochhammer}\left(\frac{1}{2}\,a+\frac{1}{2}\,b, N\right)\mathrm{pochhammer}\left(-\frac{1}{2}+\frac{1}{2}\,b+\frac{1}{2}\,a, N\right)N!\right)$$

Using Koepf Exercise 1.3, this simplifies to the following (I found no built-in routine that will do this automatically; this was done by hand):

```
> s_even2 := (-2)^(n/2) * (n-1)!/(2^((n-2)/2)*((n-2)/2)!)
  * pochhammer(a+b+n-1,n/2) /
  (pochhammer(a,n/2)*pochhammer(b,n/2));
```

$$s\_even2 := \frac{(\text{-}2)^{(1/2\,n)}\,(n-1)!\,\mathrm{pochhammer}\left(n-1+a+b, \frac{1}{2}\,n\right)}{2^{(1/2\,n-1)}\left(\frac{1}{2}\,n-1\right)!\,\mathrm{pochhammer}\left(a, \frac{1}{2}\,n\right)\mathrm{pochhammer}\left(b, \frac{1}{2}\,n\right)}$$

Verify that s_even, s_even2, agree:

```
'simplify(subs(N=nn,s_even)-subs(n=2*nn,s_even2))'$'nn'=1.
.5;
```

$$0, 0, 0, 0, 0$$

Similarly for odd terms, do n=2N+1.

```
> s_odd:=closedform(subs(n=2*N+1,Fnk),k,N);
```

$$s\_odd := N!\,\mathrm{pochhammer}\left(\frac{1}{3}\,a+\frac{1}{2}+\frac{1}{3}\,b, N\right)\mathrm{pochhammer}\left(\frac{1}{6}+\frac{1}{3}\,b+\frac{1}{3}\,a, N\right)$$

$$\text{pochhammer}\left(\frac{1}{3}a+\frac{5}{6}+\frac{1}{3}b, N\right)(-27)^N \Big/ \left(\text{pochhammer}\left(b+\frac{1}{2}, N\right)\right.$$
$$\left.\text{pochhammer}\left(a+\frac{1}{2}, N\right)\text{pochhammer}\left(\frac{1}{2}a+\frac{1}{2}+\frac{1}{2}b, N\right)\text{pochhammer}\left(\frac{1}{2}a+\frac{1}{2}b, N\right)\right)$$

For odd n, there is a catch --- this answer is absolutely wrong. It should be identically 0. There is a singularity of some sort that the program didn't catch.
The series is terminating because the upper parameter -n results in pochhammer(-n,k)=0 for k>=n, so let's compute it directly by summing Fnk(n,k) over k=0..n:

```
>
> Fn_ := proc(nn)
    local kk;
    factor(sum('simpcomb(subs(n=nn,k=kk,Fnk))',
  kk=0..nn));
  end:
```

and now evaluate it for n from 0 to 10:

```
> for nn from 0 to 10 do FN[nn] := Fn_(nn) od;
```

$$FN_0 := undefined$$

$$FN_1 := 0$$

$$FN_2 := -2\,\frac{b+1+a}{a\,b}$$

$$FN_3 := 0$$

$$FN_4 := 12\,\frac{(a+b+4)\,(a+b+3)}{(1+a)\,a\,b\,(1+b)}$$

$$FN_5 := 0$$

$$FN_6 := -120\,\frac{(a+7+b)\,(a+6+b)\,(a+5+b)}{a\,(1+a)\,(a+2)\,b\,(1+b)\,(b+2)}$$

$$FN_7 := 0$$

$$FN_8 := 1680\,\frac{(a+10+b)\,(a+9+b)\,(a+8+b)\,(a+7+b)}{a\,(1+a)\,(a+2)\,(3+a)\,b\,(1+b)\,(b+2)\,(3+b)}$$

$$FN_9 := 0$$

$$FN_{10} := -30240\,\frac{(a+9+b)\,(a+13+b)\,(a+12+b)\,(a+11+b)\,(a+10+b)}{(1+a)\,(a+2)\,(3+a)\,(a+4)\,(1+b)\,(b+2)\,(3+b)\,(b+4)\,b\,a}$$

Empirically, when n is odd, the sum is 0. To prove it, we have the recursion above relating s(n) to s(n+2); since it's 0 for n=1, it's 0 for n=3,5,7,9,... by iterating the recursion. Now we consider even n.

```
> lcoeff(FN[10]);
```

$$-30240$$

```
> 'ifactor(lcoeff(FN[2*nn+2])/lcoeff(FN[2*nn]))'$'nn'=0..4
  ;
```

$$-(2), -(2)\,(3), -(2)\,(5), -(2)\,(7), -(2)\,(3)^2$$

Empirically, when n=2N is even, it is given by the formula above called s_even2.

Now check whether this direct computation agrees with the automatically discovered formula s_even:

```
> 'simplify(FN[2*nn]-subs(N=nn,s_even))'$'nn'=1..5;
```

$$0, 0, 0, 0, 0$$

```
>
```

Koepf 7.21

This is like the handout for Koepf problem 4.7 done in class 10/29/99. It's almost the same except for the functions plugged in.

```
> sumrecursion(binomial(n-k,k)*x^k,k=0..n/2,s(n));
```

$$-\mathrm{s}(n+2)+\mathrm{s}(n+1)+x\,\mathrm{s}(n)=0$$

```
> sumrecursion(binomial(n+1,2*k+1)*(1+4*x)^k / 2^n,
  k=0..n/2, s(n));
```

$$-\mathrm{s}(n+2)+\mathrm{s}(n+1)+x\,\mathrm{s}(n)=0$$

The recursions are the same! Check initial conditions.

```
> f1 := n -> sum(binomial(n-k,k)*x^k,k=0..n/2);
  f2 := n -> sum(binomial(n+1,2*k+1)*(1+4*x)^k / 2^n,
  k=0..n/2);
```

$$f1 := n \to \sum_{k=0}^{1/2\,n} \mathrm{binomial}(n-k, k)\, x^k$$

$$f2 := n \to \sum_{k=0}^{1/2\,n} \frac{\mathrm{binomial}(n+1, 2\,k+1)\,(1+4\,x)^k}{2^n}$$

```
> f1(0)=f2(0), f1(1)=f2(1);
```

$$1=1, 1=1$$

That's all that's necessary. Check it for more. The variable nn was already set, so we must unset it.

```
> nn;
```

$$11$$

```
> nn := 'nn';
```

$$nn := nn$$

```
> nn;
```

$$nn$$

```
> 'f1(nn)=f2(nn)'$nn=0..5;
```

$$1=1,\ 1=1,\ 1+x=1+x,\ 1+2\,x=1+2\,x,\ 1+3\,x+x^2=\frac{15}{16}+\frac{5}{2}x+\frac{1}{16}(1+4\,x)^2,$$
$$1+4\,x+3\,x^2=\frac{13}{16}+\frac{5}{2}x+\frac{3}{16}(1+4\,x)^2$$

```
> 'f1(nn)=expand(f2(nn))'$nn=0..5;
```

$$1=1,\ 1=1,\ 1+x=1+x,\ 1+2\,x=1+2\,x,\ 1+3\,x+x^2=1+3\,x+x^2,$$
$$1+4\,x+3\,x^2=1+4\,x+3\,x^2$$

Fibonacci numbers:
This sum at x=1 gives the Fibonacci numbers. Thus,

```
> Sumtohyper(binomial(n-k,k)*x^k,k);
```

$$\mathrm{Hypergeom}\left(\left[-\frac{1}{2}n, \frac{1}{2}-\frac{1}{2}n\right], [-n], -4\,x\right)$$

```
> g1 := subs(Hypergeom=hypergeom,x=1,");
```

$$g1 := \mathrm{hypergeom}\left(\left[-\frac{1}{2}n, \frac{1}{2}-\frac{1}{2}n\right], [-n], -4\right)$$

```
> 'evalf(subs(n=nn,g1))'$nn=1..6;
```

1.000000000, 2.000000000, 3.000000000, 5.000000000, 8.000000000,
13.00000000

```
> Sumtohyper(binomial(n+1,2*k+1)*(1+4*x)^k / 2^n, k);
```

$$(n+1)\,2^{(-n)}\,\mathrm{Hypergeom}\left(\left[-\frac{1}{2}n, \frac{1}{2}-\frac{1}{2}n\right], \left[\frac{3}{2}\right], 1+4\,x\right)$$

```
> g2 := subs(Hypergeom=hypergeom,x=1,");
```

$$g2 := (n+1)\,2^{(-n)}\,\mathrm{hypergeom}\left(\left[-\frac{1}{2}n, \frac{1}{2}-\frac{1}{2}n\right], \left[\frac{3}{2}\right], 5\right)$$

```
> 'evalf(subs(n=nn,g2))'$nn=1..6;
```

1.000000000, 2.000000000, 3.000000000, 5.000000000, 8.000000001,
13.00000000

```
>
```

Koepf 7.24(b)

We may write the product as

```
> Sum(Sum(Hyperterm([a],[b],x,k)*Hyperterm([a],[b],-x,m),k
```

```
=0..infinity),m=0..infinity);
```

$$\sum_{m=0}^{\infty}\left(\sum_{k=0}^{\infty}\text{Hyperterm}([a],[b],x,k)\,\text{Hyperterm}([a],[b],-x,m)\right)$$

and letting n = m+k, we pull out the factor x^n to get

```
> Sum(Sum(Hyperterm([a],[b],1,k)*Hyperterm([a],[b],-1,n-k)
  ,k=0..n)*x^n,n=0..infinity);
```

$$\sum_{n=0}^{\infty}\left(\sum_{k=0}^{n}\text{Hyperterm}([a],[b],1,k)\,\text{Hyperterm}([a],[b],-1,n-k)\right)x^n$$

Now find an expression for the inside summation.

```
> sumrecursion(hyperterm([a],[b],1,k)*hyperterm([a],[b],-1
  ,n-k),k=0..n,s(n));
```

$$-(n+2)(n+2b)(b+n+1)(b+n)\,\text{s}(n+2)+(2a+n)(-2a+2b+n)\,\text{s}(n)=0$$

Once again, odd and even are different cases! We can tell from the definition of (b) that it is an even function of x, so the odd terms all have coefficient 0. Thus, we rewrite the product again using only even exponents:

```
> Sum(Sum(Hyperterm([a],[b],1,k)*Hyperterm([a],[b],-1,2*n-
  k),k=0..2*n)*x^(2*n),n=0..infinity);
```

$$\sum_{n=0}^{\infty}\left(\sum_{k=0}^{2n}\text{Hyperterm}([a],[b],1,k)\,\text{Hyperterm}([a],[b],-1,2n-k)\right)x^{(2n)}$$

```
> sumrecursion(hyperterm([a],[b],1,k)*hyperterm([a],[b],-1
  ,2*n-k),k,s(n));
```

$$-(n+1)(2n+b)(b+n)(2n+1+b)\,\text{s}(n+1)+(n+a)(-a+b+n)\,\text{s}(n)=0$$

That’s better, now it will be able to solve it.

```
> insidesum :=
  closedform(hyperterm([a],[b],1,k)*hyperterm([a],[b],-1,2
  *n-k),k,n) * x^(2*n);
```

$$insidesum := \frac{\text{pochhammer}(a,n)\,\text{pochhammer}(-a+b,n)\left(\frac{1}{4}\right)^n x^{(2n)}}{\text{pochhammer}\left(\frac{1}{2}b,n\right)\text{pochhammer}(b,n)\,\text{pochhammer}\left(\frac{1}{2}+\frac{1}{2}b,n\right)n!}$$

```
> Sumtohyper(insidesum,n);
```

$$\text{Hypergeom}\left([a,-a+b],\left[\frac{1}{2}+\frac{1}{2}b,\frac{1}{2}b,b\right],\frac{1}{4}x^2\right)$$

```
>
```